ВЫБРАТЬ Ссылка КАК Характеристика, ТипЗначения(Значение) КАК ТипЗначенияРеквизита
ИЗ Справочник.ХарактеристикиНоменклатуры.ДополнительныеРеквизиты
ГДЕ Свойство.Наименование = "КРМ"

выдаёт, что ТипЗначенияРеквизита везде число. при попытке произвести со Значение какие-нибудь арифметические операции выскакивает ошибка "неверные параметры". если сделать ВЫРАЗИТЬ(Значение КАК ЧИСЛО(10,3)), то происходят разные вещи: в консоли запрос начинает выполняться, а в расширении конфигурации - выскакивает ошибка "несовместимые типы".

как заставить выполняться запрос в расширении?
---
правьте макет отчета не в расширении. ошибки в конструкторе не влияют на работу отчета.
---
Если не хочется делать максимально изысканно, можно постобработкой получить типы значений.
---
не понял, что значит "максимально изысканно" и "постобработка". делаю отчёт в СКД.
---
Понял Вас, это важное уточнение: делаю отчёт в СКД

Тогда мне нужна задача: для чего Вы это делаете и какой ожидаемый результат. Без внятного ТЗ, как говорится. :)
---
в дальнейшем я использую эти коэффициенты для изменения количества материала, расходуемого в соответствии с ресурсными спецификациями:

```1c
    ВЫБРАТЬ
    ХарактеристикиНоменклатурыДополнительныеРеквизиты.Ссылка КАК Размер,
    ВЫРАЗИТЬ(ХарактеристикиНоменклатурыДополнительныеРеквизиты.Значение КАК ЧИСЛО(10, 3)) КАК Коэффициент
    ПОМЕСТИТЬ КРМ
    ИЗ
    Справочник.ХарактеристикиНоменклатуры.ДополнительныеРеквизиты КАК ХарактеристикиНоменклатурыДополнительныеРеквизиты
    ГДЕ
    ХарактеристикиНоменклатурыДополнительныеРеквизиты.Свойство.Наименование = "КРМ"
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Материалы.Ссылка КАК Спецификация,
    Изделие.Номенклатура КАК НоменклатураИзделия,
    Изделие.Характеристика КАК ХарактеристикаИзделия,
    ВЫБОР
    КОГДА Изделие.Упаковка = ЗНАЧЕНИЕ(Справочник.УпаковкиЕдиницыИзмерения.ПустаяСсылка)
    ТОГДА Изделие.КоличествоУпаковок
    ИНАЧЕ Изделие.КоличествоУпаковок * Изделие.Упаковка.Числитель
    КОНЕЦ КАК КоличествоИзделия,
    Материалы.Номенклатура КАК НоменклатураМатериала,
    Материалы.Характеристика КАК ХарактеристикаМатериала,
    ВЫБОР
    КОГДА Материалы.Упаковка = ЗНАЧЕНИЕ(Справочник.УпаковкиЕдиницыИзмерения.ПустаяСсылка)
    ТОГДА Материалы.КоличествоУпаковок * ЕСТЬNULL(КРМ.Коэффициент, 1)
    ИНАЧЕ Материалы.КоличествоУпаковок * Материалы.Упаковка.Числитель * ЕСТЬNULL(КРМ.Коэффициент, 1)
    КОНЕЦ КАК КоличествоМатериала
    ПОМЕСТИТЬ НормыРасхода
    ИЗ
    Справочник.РесурсныеСпецификации.МатериалыИУслуги КАК Материалы
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.РесурсныеСпецификации.ВыходныеИзделия КАК Изделие
    ПО (Изделие.Ссылка = Материалы.Ссылка)
    И (Изделие.НомерСтроки = 1)
    ЛЕВОЕ СОЕДИНЕНИЕ КРМ КАК КРМ
    ПО Материалы.Характеристика = КРМ.Размер
    ГДЕ
    Материалы.Ссылка В
    (ВЫБРАТЬ
    РегистрСведений.ОсновныеСпецификации.Спецификация
    ИЗ
    РегистрСведений.ОсновныеСпецификации)
```

эта часть запроса нормально выполняется во внешнем отчёте, но, когда мне нужно то же сделать в отчёте расширения конфигурации, то получается то, о чём я писал в первом сообщении.
---
вы из внешнего отчета перенесли код в расширение и ошибка идет при отладке или уже при исполнении в режиме предприятия?
---
ошибки выходят при попытке записать изменения в запросе на вкладке "Наборы данных".
---
как правильно сказали в посте (8) текст запроса необходимо отлаживать во внешнем отчете. В расширение переносить уже готовый запрос, запрос можно изменить вручную, не конструктором запроса
---
во внешнем отчёте выскакивает ошибка "Таблица не найдена" относительно регистра сведений, имеющегося только в расширении.
---
правьте макет отчета не в расширении. ошибки в конструкторе не влияют на работу отчета.
---
единственное место, которое я нашёл, где конструктор запросов позволяет работать и с дополнительными реквизитами характеристик, выраженными как число, и с объектами из расширения конфигурации, - это консоль СКД в пользовательском режиме. оттуда сохраняю схему в XML-файл, в отчёте расширения загружаю схему и записываю изменения - отчёт работает.
---
Немножко некропост, но не увидел здесь ответа, поэтому напишу.

Приведение типов для дополнительных реквизитов в запросе работает криво, следующая строчка выдаст ошибку несовместимых типов:

ВЫРАЗИТЬ(НоменклатураДополнительныеРеквизиты.Значение КАК Справочник.Номенклатура)

А эта строчка пройдёт:

ВЫРАЗИТЬ(ЕСТЬNULL(НоменклатураДополнительныеРеквизиты.Значение, ЗНАЧЕНИЕ(Справочник.Номенклатура.ПустаяСсылка)) КАК Справочник.Номенклатура)

Вывод: если работаете с дополнительными реквизитами в запросе и требуется явное приведение типа (что хорошо делать, чтобы не получать лютосоставной тип колонки), то это можно сделать через конструкцию ЕСТЬNULL и указание в паре пустого значения нужного типа.